

Aplicaciones del cálculo integral

El objetivo de esta práctica es usar Maple para hallar integrales definidas y estudiar aplicaciones del cálculo integral: cálculo de sumas de Riemann, áreas, volúmenes de revolución, longitudes, etc.

1. Comentarios generales

Para escribir y calcular integrales definidas utilizamos los siguientes comandos:

> `Int(x^2+2*x-3, x=1..5);`

$$\int_1^5 (x^2 + 2x - 3) dx$$

El comando `value(Int(...))` o directamente `int` nos dan el valor:

> `Int(x^2+2*x-3, x=1..5)=int(x^2+2*x-3, x=1..5);`

$$\int_1^5 (x^2 + 2x - 3) dx = \frac{160}{3}$$

Para hacer cambios de variable, necesitamos cargar el paquete `student`:

> `with(student):a:=Int(1/(4*x+17)^6, x=1..2);`
`al:=changevar(u=4*x+17, a, u);value(al);`

$$a := \int_1^2 \frac{1}{(4x+17)^6} dx$$

$$al := \int_{21}^{25} \frac{1}{4u^6} du$$


$$\frac{1420381}{199418994140625}$$

2. Sumas de Riemann

Vamos a dibujar y calcular algunas sumas de Riemann de $f(x) = x^3 - 5x^2 + x + 10$ en el intervalo $[-1, 4]$. Para eso, introducimos el paquete `student` y dibujamos la curva.


> `restart; with(student):f:=x->x^3-5*x^2+x+10;`

> `plot(f(x), x=-1..4, thickness=2);`



Vamos a dividir el intervalo en subintervalos; dibujamos un rectángulo sobre cada subintervalo de altura igual al valor de f en el extremo derecho del subintervalo. Para ello usamos el comando `rightbox` del paquete `student`. Calculamos el valor del área encerrada y su aproximación (cambiando `right` por `left` o `middle` podemos considerar rectángulos de altura igual al valor de f en el extremo izquierdo o en el punto medio).

> `rightbox(f(x), x=-1..4, 6);rightsum(f(x), x=-1..4, 6);evalf(%);`




$$\frac{5}{6} \sum_{i=1}^6 \left(\left(-1 + \frac{5}{6} i \right)^3 - 5 \left(-1 + \frac{5}{6} i \right)^2 + 9 + \frac{5}{6} i \right)$$

$$10.54398148$$

Ahora con 25 rectángulos:

> `rightbox(f(x), x=-1..4, 25);rightsum(f(x), x=-1..4, 25);evalf(%);`



$$12.40000000$$

El valor de la integral es el siguiente:


```
> Int(f(x), x=-1..4)=int(f(x), x=-1..4);evalf(int(f(x), x=-1..4));
```

$$\int_{-1}^4 (x^3 - 5x^2 + x + 10) dx = \frac{155}{12}$$

12.91666667

Programemos este proceso de aproximación, dándole animación:

```
> animateRiemann:=proc(f,a,b)
  local j,Lb;
  for j from 1 to 10 do
    Lb[j]:=student[rightbox](f,x=a..b,20*j):
  end do;
  plots[display]([seq(Lb[j], j=1..10)], insequence=true);
end proc;
> animateRiemann(f,-1,4);
```



Calculemos ahora algunas aproximaciones numéricas:

```
> for j from 1 by 5 to 26 do
  print(evalf[10](rightsum(f(x), x=-1..4, 300*j)))
end do;
```

12.87488426

12.90971901

12.91287783

12.91406206

12.91468228

12.91506393

3. Ejercicios resueltos

1) Calcular $\int_2^3 \frac{x^2}{x^4-1} dx$. Calcular $\int_0^a \frac{x^2}{x^4-1} dx$, si $0 < a < 1$.

```
> restart: with(student): int(x^2/(x^4-1), x);
```

$$\frac{1}{4} \ln(x-1) - \frac{1}{4} \ln(x+1) + \frac{1}{2} \arctan(x)$$

```
> Int(x^2/(x^4-1), x =2..3)=int(x^2/(x^4-1), x=2..3);
```

$$\int_2^3 \frac{x^2}{x^4-1} dx = \frac{1}{4} \ln(3) - \frac{1}{2} \arctan(2) - \frac{1}{4} \ln(2) + \frac{1}{2} \arctan(3)$$

```
> int(x^2/(x^4-1), x=0..a);
```

Warning, unable to determine if -1 is between 0 and a; try to use assumptions or use the AllSolutions option

Warning, unable to determine if 1 is between 0 and a; try to use assumptions or use the AllSolutions option

$$\int_0^a \frac{x^2}{x^4-1} dx$$

Siguiendo los consejos anteriores:

```
> int(x^2/(x^4-1), x=0..a, AllSolutions);
```

$$\left\{ \begin{array}{ll} \text{undefined} & a < -1 \\ \infty & a = -1 \\ \frac{1}{4} \ln(1-a) - \frac{1}{4} \ln(a+1) + \frac{1}{2} \arctan(a) & a \leq 0 \\ -\frac{1}{4} i\pi + \frac{1}{4} \ln(a-1) - \frac{1}{4} \ln(a+1) + \frac{1}{2} \arctan(a) & a < 1 \\ -\infty & a = 1 \\ \text{undefined} & 1 < a \end{array} \right.$$

Podemos usar una primitiva válida a la izquierda de 1 y obtener el resultado:

```
> f:=x->1/4*ln(abs(x-1))-1/4*ln(abs(1+x))+1/2*arctan(x);
Int(x^2/(x^4-1), x=0..a)=f(a)-f(0);
```

$$f := x \mapsto \frac{1}{4} \ln(|x-1|) - \frac{1}{4} \ln(|x+1|) + \frac{1}{2} \arctan(x)$$

$$\int_0^a \frac{x^2}{x^4-1} dx = \frac{1}{4} \ln(|a-1|) - \frac{1}{4} \ln(|a+1|) + \frac{1}{2} \arctan(a)$$

Maple sí sabe hallar la integral si le damos más datos sobre a:

```
> assume(0<a,a<1);int(x^2/(x^4-1), x=0..a);
normal(%-(f(a)-f(0)));
```

$$\frac{1}{4} \ln(1-a) - \frac{1}{4} \ln(a+1) + \frac{1}{2} \arctan(a)$$


0

2) Sean $g(x) = x^2 + 2$ y $f(x) = 2x + 5$. Dibujar las gráficas de las dos funciones sobre los mismos ejes y encontrar el área encerrada entre las dos curvas y las rectas $x = 0$ y $x = 6$.

```
> restart: with(plots):
> g:=x->x^2+2;
> f:=x->2*x+5;
> plot([g(x),f(x)],x=0..6,thickness=2,color=[red,blue]);
```

$$g := x \rightarrow x^2 + 2$$

$$f := x \rightarrow 2x + 5$$



El área de la figura comprendida entre las dos curvas es:

```
> Int('abs(f(x)-g(x))',x=0..6)=int(abs(f(x)-g(x)),x=0..6);
```

$$\int_0^6 |f(x) - g(x)| dx = 36$$

Si queremos calcular la integral a mano, es conveniente estudiar el signo de $f(x) - g(x)$, para quitar el valor absoluto:

```
> solve(f(x)=g(x),x);
```

$$-1, 3$$

Por lo tanto, el área es:

```
> Int('abs(f(x)-g(x))',x=0..6)=
Int('f(x)-g(x)',x=0..3)+Int('g(x)-f(x)',x=3..6);
```

$$\int_0^6 |f(x) - g(x)| dx = \int_0^3 (f(x) - g(x)) dx + \int_3^6 (g(x) - f(x)) dx$$

Comprobemos que, en efecto, el resultado es el mismo:

```
> int(f(x)-g(x),x=0..3)+int(g(x)-f(x),x=3..6);
```

$$36$$

3) ¿Para qué valores de m encierran una región la recta $y = mx$ y la curva $y = \frac{x}{x^2 + 1}$?


Encontrar su área.

```
> restart: with(plots):
> f:=x->x/(x^2+1):g:=x->m*x;
```

No se puede representar la función g sin dar un valor concreto a la constante m , así que elegimos un valor para hacernos una idea:

```
> m:=1/4;
> plot([f(x),g(x)],x=-2..2,thickness=2,color=[blue,red]);
```

$$m := \frac{1}{4}$$



Hallemos los puntos de corte. Pero debemos hacerlo en general, y no para el valor concreto de m elegido antes. Así que primero borramos el valor de m con la orden **unassign**:

```
> unassign('m');
> solve({f(x)=g(x)},x);
```

$$\{x=0\}, \left\{x = \frac{\sqrt{-m(-1+m)}}{m}\right\}, \left\{x = -\frac{\sqrt{-m(-1+m)}}{m}\right\}$$

Las soluciones no nulas son reales si y solo si $0 < m < 1$. El área de la superficie es:

```
> assume(0<m,m<1):
> int(abs(f(x)-g(x)),
> x=-1/m*(-m*(-1+m))^(1/2)..1/m*(-m*(-1+m))^(1/2));
-ln(m~) - 1 + m~
```

O de otra forma; por simetría, el área es:


```
> 2*int(f(x)-g(x),x=0..sqrt(-m*(-1+m))/m);
-ln(m~) - 1 + m~
```

4) Encontrar el volumen de revolución engendrado por rotación alrededor del eje OX de la región del plano limitada por las curvas $y = \sec(x)$, $y = 1/2$, $x = -1$, $x = 1$. Dar un dibujo de la región del plano y de la sección del volumen originado.

```
> restart; with(plots):
```


Representemos la región:

```
> secante:=plot(sec(x),x=-1..1,color=blue):
recta:=plot(1/2,x=-1..1,color=red,thickness=2):
vert1:=plot([-1,t,t=1/2..sec(-1)],color=black):
vert2:=plot([1,t,t=1/2..sec(1)],color=black):
display(secante,recta,vert1,vert2,view=[-1..1,0..sec(1)]);
```



Aunque no es imprescindible, podemos dibujar ahora el sólido generado. Cuando una curva $y = f(x)$ gira alrededor del eje de abscisas, genera una superficie de ecuación $x^2 + z^2 = f(y)^2$.

```
> exterior:=implicitplot3d(x^2+z^2=sec(y)^2,
x=-2..2,y=-1..1,z=-2..2):
interior:=implicitplot3d(x^2+z^2=1/4,
x=-1/2..1/2,y=-1..1,z=-1/2..1/2):
display(exterior,interior,scaling=constrained,
orientation=[45,-71],lightmodel=light2,axes=boxed);
```



Calculemos el volumen:

```
> V:=int(Pi*sec(x)^2,x=-1..1)-int(Pi/4,x=-1..1);
V:= \frac{2 \sin(1) \pi}{\cos(1)} - \frac{1}{2} \pi
```

5) Calcular el volumen engendrado por la rotación alrededor del eje OY de la parte del plano limitada por las curvas $y = 5(x^2 - x^3)$, $y = 0$.

```
> restart; with(plots):
g:=x->5*(x^2-x^3);
```

$$g := x \rightarrow 5x^2 - 5x^3$$


Para saber de qué parte del plano se habla, calculamos los puntos de corte de la curva $y = 5(x^2 - x^3)$ y la recta $y = 0$:

```
> solve({g(x)=0});
```

$$\{x=0\}, \{x=0\}, \{x=1\}$$


Dibujemos la gráfica en el dominio correspondiente para girarla alrededor de OY

```
> plot(g(x),x=-0..1,scaling=constrained);
```



La superficie que genera la curva $y = g(x)$ al girar alrededor del eje de ordenadas tiene la ecuación $z = g(\sqrt{x^2 + y^2})$. Podemos representarla:

```
> plot3d(g(sqrt(x^2+y^2)),x=-1..1,y=-sqrt(1-x^2)..sqrt(1-x^2),
scaling=constrained,orientation=[90,60],axes=none);
```



Calculemos el volumen:

```
> V:=int(2*Pi*x*g(x),x=0..1)=int(2*Pi*x*g(x),x=0..1);
V:= \int_0^1 2 \pi x (5x^2 - 5x^3) dx = \frac{1}{2} \pi
```